```
> restart;
> with(LinearAlgebra):
> m:=4;
```

$$m := 4$$

```
> A:=Matrix([[0,0],[1/3,1/3]]);
```

$$A := \begin{bmatrix} 0 & 0 \\ \frac{1}{3} & \frac{1}{3} \end{bmatrix}$$

```
> b:=Vector([1/4,3/4]);
```

$$b := \begin{bmatrix} \frac{1}{4} \\ \frac{3}{4} \end{bmatrix}$$

```
> c:=Vector([0,2/3]);
```

$$c := \begin{bmatrix} 0 \\ \frac{2}{3} \end{bmatrix}$$

```
> n:=Dimension(b);
```

$$n := 2$$

```
> onesvector:=Vector(n,1);
  fxi:=Vector([seq(f(t+c[i]*h,xi[i]),i=1..n)]);
  xirhs:=y(t)*onesvector+h*Multiply(A,fxi);
```

$$onesvector := \begin{bmatrix} 1 \\ 1 \end{bmatrix}$$

$$fxi := \begin{bmatrix} f(t, \xi_1) \\ f\left(t + \frac{2}{3} h, \xi_2 \right) \end{bmatrix}$$

$$xirhs := \begin{bmatrix} y(t) \\ y(t) + h\left(\frac{1}{3} f(t, \xi_1) + \frac{1}{3} f\left(t + \frac{2}{3} h, \xi_2 \right) \right) \end{bmatrix}$$

```
> T:=y(t+h)-y(t)-h*Multiply(Transpose(b),fxi);
```

```
for j from 1 to m-1
do
    T:=subs(xi=xirhs,T);
od:
```

$$T := y(t + h) - y(t) - h\left(\frac{1}{4} f(t, \xi_1) + \frac{3}{4} f\left(t + \frac{2}{3} h, \xi_2\right)\right)$$

```
> S:=series(T,h,m);
```

$$S := (-f(t, y(t)) + \mathrm{D}(y)(t))\, h + \left(-\frac{1}{2}\mathrm{D}_1(f)(t, y(t)) - \frac{1}{2}\mathrm{D}_2(f)(t, y(t))\, f(t, y(t)) + \frac{1}{2}\mathrm{D}^{(2)}(y)(t)\right) h^2 +$$

$$-\frac{1}{6}\mathrm{D}_2(f)(t, y(t))^2\, f(t, y(t)) - \frac{1}{6} f(t, y(t))^2\, \mathrm{D}_{2,2}(f)(t, y(t)) - \frac{1}{6}\mathrm{D}_2(f)(t, y(t))\, \mathrm{D}_1(f)(t, y(t))$$

$$-\frac{1}{6}\mathrm{D}_{1,1}(f)(t, y(t)) - \frac{1}{3}\mathrm{D}_{1,2}(f)(t, y(t))\, f(t, y(t)) + \frac{1}{6}\mathrm{D}^{(3)}(y)(t)\Big) h^3 + O(h^4)$$

```
> eq[1]:=D(t->y(t))(t)=f(t,y(t));
```

$$eq_1 := \mathrm{D}(y)(t) = f(t, y(t))$$

```
> for j from 1 to m-2
  do
      eq[j+1]:=simplify(subs(seq(eq[i],i=1..j),D(unapply(eq[j],t))(t)));
  od;
```

$$eq_2 := \mathrm{D}^{(2)}(y)(t) = \mathrm{D}_1(f)(t, y(t)) + \mathrm{D}_2(f)(t, y(t))\, f(t, y(t))$$

$$eq_3 := \mathrm{D}^{(3)}(y)(t) = \mathrm{D}_{1,1}(f)(t, y(t)) + 2\,\mathrm{D}_{1,2}(f)(t, y(t))\, f(t, y(t)) + f(t, y(t))^2\, \mathrm{D}_{2,2}(f)(t, y(t))$$

$$+ \mathrm{D}_2(f)(t, y(t))\, \mathrm{D}_1(f)(t, y(t)) + \mathrm{D}_2(f)(t, y(t))^2\, f(t, y(t))$$

```
> simplify(subs(seq(eq[i],i=1..m-1),S));
```

$$O(h^4)$$